HEWLETT PACKARD

CALCULATOR SUPPORT
1000 N.E. Circle Boulevard, Corvallis, Oregon 97330-4239, Telephone 503 757-2004 8:00 AM - 3:00 PM Pacific Time

HP 12C SOLUTIONS

The following solutions have been developed as a continuing effort by Hewlett-Packard to meet the needs of our customers.

TABLE OF CONTENTS

12-8 Actuarial Calculations

Collection of programs to calculate the compound interest functions n, i, PV, PMT, and FV using the actuarial method. This means that the value for n can be any positive number. The results duplicate those produced by the HP-37E, HP-38E/C, and HP-18C calculators.

12-9 Increasing Annuities

Procedure to calculate the present and future values of an annuity that increases at a constant rate and at equal intervals of time. Routines are included for both END and BEGIN mode calculations.

12-10 Annual Coupon Bond Price and Yield

Program to calculate the price, accrued interest, and yield of an annual coupon bond. The redemption can be at maturity or at call, and either calendar basis can be accommodated (30/360 or Actual/Actual).

12-11 Deposits Needed to Meet a Future Cash Flow Need

Procedure to calculate the amount of money to deposit each period to accumulate a desired amount. A typical example is "How much should I save each month to fund my daughter's college education?"

12-12 Price and Yield to Call (Actual/Actual Day Basis)

Program to calculate price and yield to call, assuming a semi-annual coupon payment and an Actual/Actual day basis.

HEWLETT PACKARD

HP-12C Solutions 12-1

Reverse Annuity Mortgage Payment

Reverse Annuity Mortgage

A reverse annuity mortgage allows people over 62 years of age to use the equity they have in their homes to generate regular monthly income. The following procedure determines the amount of the monthly payment that they will receive.

1. Press [g] [BEG] and [f] [FIN].
2. Key in the number of years in the loan and press [g] [12×].
3. Key in the annual interest rate (as a percentage) and press [g] [12÷].
4. Key in the amount of the initial cash flow (if it exists) and press [PV].
5. Key in the total loan amount and press [CHS] [FV].
6. Press [PMT] to calculate the monthly payment.

Example.
Loan amount = $64000
Term = 5 years
Interest rate = 13%
Initial payment = $2500

Keys:	Display:	Description:
[g] [BEG]		Selects Begin mode.
[f] [FIN]		Clears finance registers.
5 [g] [12×]	60.00	Stores known values.
13 [g] [12÷]	1.08	
2500 [PV]	2,500.00	
64000 [CHS] [FV]	-64,000.00	
[PMT]	698.41	Calculates monthly payment received.

Graduated Reverse Annuity Mortgage

A graduated reverse annuity mortgage means that the monthly payments are less in the beginning, and increase at a specified rate for the term of the loan. A program to calculate the monthly payments follows. Key in the program, then use this procedure.

1. Key in the number of years in the loan and press [STO] 0.
2. Key in the annual interest rate (as a percentage) and press [STO] 1.
3. Key in the total loan amount and press [STO] 2.
4. Key in the annual percentage increase in the payment and press [STO] 3.
5. Key in the amount of the initial cash flow and press [STO] 4. If there is no initial cash flow, key in 0.
6. Press [R/S] to calculate the payment received in the first year.
7. Repeat step 6 to calculate the payment for subsequent years.
8. Optional: Press [RCL] 5 to display the loan factor. Press [RCL] 6 to display the initial cash flow factor.
9. For a new loan, return to step 1 and key in any values that have changed.

Example.
Loan amount = $64000
Term = 5 years
Interest rate = 13%
Initial payment = $2500
Increase in payment = 6%

If you have not keyed in the program, do it now.

Keys:	Display:	Description:
5 [STO] 0	5.00	Stores known values.
13 [STO] 1	13.00	
64000 [STO] 2	64,000.00	
6 [STO] 3	6.00	
2500 [STO] 4	2,500.00	
[R/S]	628.85	Calculates payment in year 1.
[R/S]	666.58	Calculates payment in year 2.
[R/S]	706.58	Calculates payment in year 3.
[R/S]	748.97	Calculates payment in year 4.
[R/S]	793.91	Calculates payment in year 5.

Graduated Reverse Annuity Mortgage Payment Program.

Keys	Display
f P/R	
f PRGM	00-
f FIN	01- 42 34
g END	02- 43 8
1	03- 1
RCL 3	04- 45 3
%	05- 25
+	06- 40
RCL 1	07- 45 1
1	08- 1
2	09- 2
0	10- 0
0	11- 0
÷	12- 10
1	13- 1
+	14- 40
1	15- 1
2	16- 2
y^x	17- 21
÷	18- 10
1	19- 1
−	20- 30
EEX	21- 26
2	22- 2
×	23- 20
1	24- 12
1	25- 1
CHS	26- 16
PMT	27- 14

Keys	Display
RCL 0	28- 45 0
STO 7	29- 44 7
n	30- 11
FV	31- 15
STO PMT	32- 44 14
g BEG	33- 43 7
RCL 1	34- 45 1
g 12÷	35- 43 12
1	36- 1
g 12×	37- 43 11
0	38- 0
FV	39- 15
PV	40- 13
RCL 0	41- 45 0
g 12×	42- 43 11
0	43- 0
PMT	44- 14
FV	45- 15
1/x	46- 22
STO 5	47- 44 5
RCL 4	48- 45 4
g x=0	49- 43 35
g GTO 63	50-43,33 63
RCL 0	51- 45 0
g 12×	52- 43 11
RCL 1	53- 45 1
g 12÷	54- 43 12
1	55- 1
PV	56- 13

Keys	Display
FV	57- 15
STO 6	58- 44 6
×	59- 20
RCL 2	60- 45 2
+	61- 40
g GTO 64	62-43,33 64
RCL 2	63- 45 2
RCL 5	64- 45 5
×	65- 20
R/S	66- 31
1	67- 1
STO − 0	68-44 30 0
RCL 0	69- 45 0
g x=0	70- 43 35
g GTO 79	71-43,33 79
R↓	72- 33
R↓	73- 33
RCL 3	74- 45 3
%	75- 25
+	76- 40
R/S	77- 31
g GTO 67	78-43,33 67
RCL 7	79- 45 7
STO 0	80- 44 0
CLX	81- 35
g GTO 00	82-43,33 00
f P/R	

Registers

n: used i: used PV: used PMT: used

FV: used R_0: # years R_1: % interest R_2: loan amount

R_3: % increase R_4: init CF R_5: loan factor R_6: CF factor

R_7: used

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-2

Annual Percentage Rate Calculation With Fees and a Balloon

This procedure calculates the APR of a loan when fees are charged (either as a percentage of the loan amount or as a flat rate) and a balloon payment is due at some time during the term of the loan. Remember to use the cash flow sign convention (money received is positive, money paid out is negative).

User Instructions.

1. Set End mode ([g] [END]) and clear the financial registers ([f] [FIN]).
2. Calculate the periodic payment amount of the loan.
   - Key in the total number of payment periods; press [n].
   - Key in the periodic interest rate; press [i].
   - Key in the mortgage amount; press [PV].
   - Press [PMT] to calculate the periodic payment amount.
3. Calculate the amount of the balloon payment.
   - Key in the number of the payment period where the balloon payment occurs; press [n].
   - Press [FV] to calculate the balloon payment amount.
4. Calculate the actual net amount disbursed.
   - If the fees are stated as a percentage of the mortgage amount (points), recall the mortgage amount ([RCL] [PV]); key in the fee percentage rate; press [%] [−] [PV].
   - If the fees are stated as a flat charge, recall the mortgage amount ([RCL] [PV]); key in the fee amount (flat charge); press [−] [PV].
   - If the fees are stated as a percentage of the mortgage amount plus a flat charge, recall the mortgage amount ([RCL] [PV]); key in the fee percentage rate; press [%] [−]; key in the fee amount (flat charge) and press [−] [PV].
5. Press [i] to calculate the percentage rate per compounding period.
6. To calculate the annual nominal percentage rate, key in the number of periods per year and press [×].

Example. A 30-year, $50000 loan at 15% interst has fees of 2 points plus $150. Assuming that monthly payments are made and that the loan is paid in full at the end of the seventh year, what is the APR?

Keys:	**Display:**	**Description:**
[g] [END]		Sets End mode.
[f] [FIN]		Clears the financial registers.
30 [g] [12×]	360.00	Stores number of months.
15 [g] [12÷]	1.25	Stores monthly interest rate.
50000 [PV]	50,000.00	Stores loan amount.
[PMT]	-632.22	Calculates monthly payment amount.
7 [g] [12×]	84.00	Stores number of payments until balloon.
[FV]	-48,937.43	Calculates balloon payment.
[RCL] [PV] 2 [%] [−] 150 [−] [PV]	48,850.00	Subtracts fees from original loan amount and restores adjusted amount.
[i]	1.30	Calculates monthly interest rate.
12 [×]	15.54	Calculates annual percentage rate.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-3

Biweekly Mortgage Payments and Amortization Schedule

One way to pay off your mortgage faster is to make biweekly mortgage payments. Instead of paying once a month, you make one-half of the monthly payment every two weeks. You make 26 or 27 payments each year (depending on the payment date), increase your equity, and pay less interest.

Example 1. Part 1. On a $75,000, 30-year mortgage at 13.5% interest, what is the amount of the biweekly payment?

Keys:	Display:	Description:
[g] [END]		Sets End mode.
[f] [FIN]		Clears the financial registers.
75000 [PV]	75,000.00	Stores loan amount.
30 [g] [12×]	360.00	Stores number of months.
13.5 [g] [12÷]	1.13	Stores monthly interest rate.
[PMT]	-859.06	Calculates monthly payment amount.
2 [÷] [PMT]	-429.53	Calculates and stores biweekly payment amount.

Part 2. With this biweekly payment, how long will it take to pay off the mortgage?

13.5 [ENTER] 26 [÷] [i]	0.52	Calculates and stores biweekly interest rate.
[n]	458.00	Calculates total number of payments.
26 [÷]	17.62	Calculates number of years.

Part 3. What is the remaining balance after 26 payments have been made?

26 [n] [FV]	-73,886.65	Calculates balance after 26 payments.

Part 4. If 27 payments are made the first year, how much interest is paid? What is the remaining balance?

0 [n]	0.00	Clears n register.
27 [f] [AMORT]	-10,438.07	Interest paid in first year.
[RCL] [PV]	73,840.76	Remaining balance.

Biweekly Amortization Schedule Program

The following program provides a biweekly amortization schedule, displaying the date of the payment, the amounts of interest and principal, and the remaining balance.

Keys:	Display:		
f P/R			
f PRGM	00-		
f FIN	01-	42	34
RCL 0	02-	45	0
g 12÷	03-	43	12
RCL 1	04-	45	1
PV	05-		13
RCL 2	06-	45	2
g 12×	07-	43	11
PMT	08-		14
f RND	09-	42	14
2	10-		2
÷	11-		10
PMT	12-		14
R/S	13-		31
0	14-		0

Keys	Display		
n	15-		11
RCL 0	16-	45	0
2	17-		2
6	18-		6
÷	19-		10
i	20-		12
1	21-		1
STO 4	22-	44	4
RCL 3	23-	45	3
1	24-		1
4	25-		4
g DATE	26-	43	16
STO 3	27-	44	3
f 6	28-	42	6
R/S	29-		31
f 2	30-	42	2

Keys	Display		
1	31-		1
f AMORT	32-	42	11
R/S	33-		31
x≷y	34-		34
R/S	35-		31
0	36-		0
RCL PV	37-	45	13
g x≤y	38-	43	34
g GTO 00	39-43,33		00
R/S	40-		31
1	41-		1
STO + 4	42-44	40	4
g GTO 23	43-43,33		23
f P/R			

User Instructions.

1. Key in the program.
2. Key in the annual interest rate as a percent and press STO 0.
3. Key in the loan amount and press STO 1.
4. Key in the original term in years and press STO 2.
5. Key in the origination date (in month, day, year format) and press STO 3.
6. Press R/S to display the amount of the biweekly payment.
7. Press R/S to display the date of the first payment.
8. Press R/S to display the interest portion of the payment.
9. Press R/S to display the principal portion of the payment.
10. Press R/S to display the remaining balance.
11. Press R/S to display the date of the next payment.
12. Return to step 8 to display the interest, principal, and remaining balance for subsequent payments.

Example 2. Generate an amortization schedule for a $60,000, 30-year, 14% mortgage, paid biweekly, that originates November 12, 1984.

Keys:	Display:	Description:
14 [STO] 0	14.00	Stores interest rate.
60000 [STO] 1	60,000.00	Stores loan amount.
30 [STO] 2	30.00	Stores number of years.
11.121984 [STO] 3	11.12	Stores origination date.
[R/S]	-355.46	Calculates biweekly payment amount.
[R/S]	11,26,1984 1	Date of first payment briefly displayed.
	11.261984	The 1 indicates a Monday.
[R/S]	-323.08	Amount of interest.
[R/S]	-32.38	Amount of principal.
[R/S]	59,967.62	Remaining balance.
[R/S]	12,10,1984 1	Date of next payment.
	12.101984	
[R/S]	-322.90	Amount of interest.
[R/S]	-32.56	Amount of principal.
[R/S]	59,935.06	Remaining balance.
[R/S]	12,24,1984 1	Date of next payment.
	12.241984	

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-4

Private Mortgage Insurance

Private mortgage insurance (PMI) is usually calculated based on a percentage of the outstanding loan balance. Since the loan balance changes, the PMI calculation is done once each year.

User Instructions.

1. For year 1:
   - Key in the loan amount and press [ENTER].
   - Key in the insurance rate as a percentage and press [%].
   - Key in the number of payments per year and press [÷].
2. For subsequent years:
   - Calculate the balance at the end of the previous year.
   - Key in the insurance rate as a percentage and press [%].
   - Key in the number of payments per year and press [÷].

Example. A 30-year, $75000 loan at 15% interest has private mortgage insurance of .25% of the loan balance. Calculate the monthly private mortgage insurance (PMI) payment for year 1.

Keys:	Display:	Description:
75000 [ENTER]	75,000.00	Enters loan amount.
.25 [%]	187.50	Calculates year 1 PMI payment.
12 [÷]	15.63	Calculates monthly PMI payment for year 1.
Calculate the monthly PMI for year 2:		
[f] [FIN]		Clears the financial registers. (The display does not change.)
30 [g] [12×]	360.00	Stores total number of payments.
15 [g] [12÷]	1.25	Stores monthly interest rate.
75000 [CHS] [PV]	-75,000.00	Stores loan amount.
[PMT]	948.33	Calculates monthly mortgage payment.
1 [g] [12×] [FV]	74,860.68	Balance at the end of the first year.
.25 [%]	187.15	Calculates year 2 PMI payment.
12 [÷]	15.60	Calculates monthly PMI payment for year 2.

Calculate the monthly PMI payment for year 3.

2 [g] [12×] [FV]	74,698.97	Balance at the end of the second year.
.25 [%]	186.75	Calculates year 3 PMI payment.
12 [÷]	15.56	Calculates monthly PMI payment for year 3.

Calculate the monthly PMI payment for year 30.

29 [g] [12×] [FV]	10,506.88	Balance at the end of year 29.
.25 [%]	26.27	Calculates year 30 PMI payment.
12 [÷]	2.19	Calculates monthly PMI payment for year 30.

Total PMI Program

The total PMI paid over a specified number of years is computed with the following program.

Keys:	Display:	Keys:	Display:	Keys:	Display:
[f] [P/R]		[STO] 2	08- 44 2	1	17- 1
[f] [PRGM]	00-	[g] [12×]	09- 43 11	[−]	18- 30
[PMT]	01- 14	[FV]	10- 15	[RCL] 2	19- 45 2
[RCL] [n]	02- 45 11	[RCL] 3	11- 45 3	[g] [x≤y]	20- 43 34
1	03- 1	[%]	12- 25	[g] [GTO] 09	21-43,33 09
2	04- 2	[STO] [+] 0	13-44 40 0	[RCL] 0	22- 45 0
[÷]	05- 10	1	14- 1	[g] [GTO] 00	23-43,33 00
[STO] 1	06- 44 1	[STO] [+] 2	15-44 40 2	[f] [P/R]	
1	07- 1	[RCL] 1	16- 45 1		

User Instructions.

1. Key in the program.
2. Press [f] [REG] to clear all registers.
3. Key in the insurance rate as a percentage and press [STO] 3.
4. Key in the loan amount and press [CHS] [PV].
5. Key in the total number of years and press [g] [12×].
6. Key in the annual interest rate as a percentage and press [g] [12÷].
7. Press [R/S] to compute the total PMI.

Example 2. Using the information in example 1, calculate the total PMI.

Keys:	**Display:**	**Description:**
f REG		Clears all registers.
.25 STO 3	0.25	Stores insurance rate.
75000 CHS PV	75,000.00	Stores loan amount.
30 g 12×	360.00	Stores total number of payments.
15 g 12÷	1.25	Stores monthly interest rate.
R/S	4,336.12	Calculates total PMI.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-5

Adjustable Rate Mortgages

An adjustable rate mortgage is a mortgage loan that provides for the adjustment of its interest rate as market interest rates change. As the interest rate changes, the amount of the periodic payment changes to reflect the new interest rate. Given the terms of the original mortgage, the changes in the interest rate, and the time frame in which the changes occur, this procedure calculates the amount of each periodic payment. Once each payment is known, the APR of the entire transaction can be calculated.

User Instructions.

1. Calculate the amount of the initial periodic payment.
2. Calculate the loan balance just before payments increase the first time, change the sign, and store the result in [PV].
3. Change the interest rate, adjust the term, store 0 in [FV], and recalculate the periodic payment.
4. Calculate the loan balance before payments increase the next time, change the sign, and store the result in [PV].
5. Repeat steps 3 and 4 until all payments are calculated.
6. Once the payments are determined, use [f][IRR] to calculate the APR.

Example 1. A $50,000, 30-year, adjustable rate mortgage has the following terms:

12% interest in first year
13% interest in second and third years
15% interest for the remaining term

What are the monthly payments?

Keys:	Display:	Description:
[f][FIN]		Clears the financial registers.
[g][END]		Sets End mode.
30 [g][12×]	360.00	Stores number of months.
12 [g][12÷]	1.00	Stores initial monthly interest rate.
50000 [CHS] [PV]	-50,000.00	Stores loan amount.
[PMT]	514.31	Calculates monthly payment amount.
12 [n] [FV] [CHS] [PV]	-49,818.56	Calculates and stores balance at end of first year.
29 [g][12×]	348.00	Stores remaining number of payments.
13 [g][12÷]	1.08	Stores new monthly interest rate.

Keys:	Display:	Description:
0 [FV]	0.00	Sets future value to zero.
[PMT]	552.70	Calculates payment amount in years two and three.
24 [n] [FV] [CHS] [PV]	-49,464.37	Calculates and stores balance at end of third year.
27 [g] [12×] 15 [g] [12÷] 0 [FV]	324,001,250.00	Stores remaining number of payments and new interest rate.
[PMT]	629.55	Calculates final payment amount.

Example 2. Calculate the APR, given the above payments.

Keys:	Display:	Description:
[f] [REG]	0.00	Clears all registers.
50000 [CHS] [g] [CFo]	-50,000.00	Stores initial cash flow.
514.31 [g] [CFj] 12 [g] [Nj] 552.70 [g] [CFj] 24 [g] [Nj] 629.55 [g] [CFj] 99 [g] [Nj] [x≷y] [g] [CFj] 99 [g] [Nj] [x≷y] [g] [CFj] 99 [g] [Nj] [x≷y] [g] [CFj] 27 [g] [Nj]		Stores subsequent cash flows. (Each group can have a maximum of 99 cash flows.)
[f] [IRR] 12 [×]	14.13	Calculates the APR.

Example 3. If the previous mortgage has a balloon payment in eight years, as well as a 2 point fee, what is the APR?

Step 1. First, calculate the balance due (the balloon payment) at the end of eight years.

Keys:	Display:	Description:
[f] [FIN]		Clears the financial registers.
49464.37 [CHS] [PV] 15 [g] [12÷] 629.55 [PMT]	-49,464.37 1.25 629.55	Store the mortgage data at the beginning of year 4.
5 [g] [12×] [FV]	48,468.32	Balance at the end of eight years.

Step 2. Calculate the APR.

[f] [REG]	0.00	Clears all registers.
50000 [CHS] [ENTER] 2 [%] [−] [g] [CFo]	-49,000.00	Stores initial cash flow.
514.31 [g] [CFj] 12 [g] [Nj] 552.70 [g] [CFj] 24 [g] [Nj] 629.55 [g] [CFj] 59 [g] [Nj]		Stores monthly cash flows.
629.55 [ENTER] 48468.32 [+] [g] [CFj]	49,097.87	Stores final cash flow.
[f] [IRR] 12 [×]	14.20	Calculates the APR.

Example 4. In the previous example, how much interest is paid in years 1, 2, 3 and 4?

Keys:	Display:	Description:
f FIN		Clears the financial registers.
50000 CHS PV	-50,000.00	Stores first year loan data.
12 g 12÷	1.00	
514.31 PMT	514.31	
12 f AMORT	5,990.23	Interest paid in year 1.
13 g 12÷	1.08	Stores second and third year data.
552.70 PMT	552.70	
12 f AMORT	6,466.76	Interest paid in year 2.
12 f AMORT	6,443.90	Interest paid in year 3.
15 g 12÷	1.25	Stores final loan data.
629.55 PMT	629.55	
12 f AMORT	7,409.96	Interest paid in year 4.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-6

Coupon Equivalent Yield

The coupon equivalent yield is a way of determining which of two investments of similar maturity will provide a higher return—a non-interest bearing obligation purchased at a discount (a Treasury Bill) or a semi-annual coupon bond on a 365-day basis (a government bond).

Program.

Keys	Display
f P/R	
f PRGM	00-
RCL 0	01- 45 0
RCL 1	02- 45 1
g ΔDYS	03- 43 26
STO 3	04- 44 3
1	05- 1
8	06- 8
2	07- 2
x≷y	08- 34
g x≤y	09- 43 34
g GTO 52	10-43, 33 52
RCL 3	11- 45 3
2	12- 2
×	13- 20
3	14- 3
6	15- 6
5	16- 5
−	17- 30
RCL 2	18- 45 2
×	19- 20
7	20- 7
3	21- 3

Keys	Display
0	22- 0
÷	23- 10
STO 4	24- 44 4
2	25- 2
×	26- 20
RCL 2	27- 45 2
+	28- 40
STO 5	29- 44 5
ENTER	30- 36
×	31- 20
RCL 2	32- 45 2
2	33- 2
×	34- 20
2	35- 2
0	36- 0
0	37- 0
−	38- 30
4	39- 4
×	40- 20
RCL 4	41- 45 4
×	42- 20
−	43- 30
g √x	44- 43 21

Keys	Display
RCL 5	45- 45 5
−	46- 30
RCL 4	47- 45 4
2	48- 2
×	49- 20
÷	50- 10
g GTO 64	51-43, 33 64
EEX	52- 26
2	53- 2
RCL 2	54- 45 2
÷	55- 10
1	56- 1
−	57- 30
3	58- 3
6	59- 6
5	60- 5
×	61- 20
RCL 3	62- 45 3
÷	63- 10
EEX	64- 26
2	65- 2
×	66- 20
f P/R	

Registers:

n: Unused
i: Unused
PV: Unused
PMT: Unused
FV: Unused
R_0: Settlement
R_1: Maturity
R_2: Price
R_3: #days
R_4: Used
R_5: Used
R_6-$R_{.0}$ Unused

User Instructions.

1. Key in the program.
2. Key in the settlement date and press [STO] 0.
3. Key in the maturity date and press [STO] 1.
4. Key in the purchase price and press [STO] 2.
5. Press [R/S] to calculate the coupon equivalent yield (as a percent).

Example. What is the coupon equivalent yield of a bond with a settlement date of July 13, 1984, a maturity date of May 1, 1985, and price of $96.78?

Keys:	Display:	Description:
[f] [REG]	0.00	Clears all registers.
7.131984 [STO] 0	7.13	Stores settlement date.
5.011985 [STO] 1	5.01	Stores maturity date.
96.78 [STO] 2	96.78	Stores bond price.
[R/S]	4.13	Calculates coupon equivalent yield.

Formulas.

- For ≤ 182 days

$$i = \left(\frac{100}{Price} - 1\right) \times \frac{365}{n}$$

- For > 182 days

$$i = \frac{-b + \sqrt{b^2 - 4ac}}{2a}$$

Where:

$$a = \frac{2\ (Price)\ (n) - (Price)\ (365)}{730}$$

$b = 2a + Price$
$c = 2\ (Price) - 200$
$n =$ actual number of days
$i =$ coupon equivalent yield (decimal)

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-7

Duration of a Bond

The duration of a bond is the average time that elapses until the various cash flows (coupons and redemption value) are received. In other words, the duration is a weighted average of the number of periods until the payments occur. For coupon-bearing bonds, the duration is always shorter than the term to maturity. (For zero-coupon bonds, duration is equal to maturity.)

Program.

Keys	Display		
f P/R			
f PRGM	00-		
PV	01-		13
RCL i	02-	45	12
EEX	03-		26
2	04-		2
÷	05-		10
STO 0	06-	44	0
1	07-		1
+	08-		40
STO 1	09-	44	1
RCL n	10-	45	11
1	11-		1
+	12-		40
y^x	13-		21

Keys	Display		
RCL 1	14-	45	1
−	15-		30
RCL 0	16-	45	0
RCL n	17-	45	11
×	18-		20
−	19-		30
RCL 1	20-	45	1
RCL n	21-	45	11
y^x	22-		21
÷	23-		10
RCL 0	24-	45	0
ENTER	25-		36
×	26-		20
÷	27-		10

Keys	Display		
RCL PMT	28-	45	14
×	29-		20
RCL FV	30-	45	15
RCL n	31-	45	11
×	32-		20
RCL 1	33-	45	1
RCL n	34-	45	11
y^x	35-		21
÷	36-		10
+	37-		40
RCL PV	38-	45	13
CHS	39-		16
÷	40-		10
f P/R			

Registers:

n: # coupons	i: Yield	PV: Used	PMT: Coupon amount
FV: Redemption	R_0: Yield/100	R_1: Yield/100 + 1	R_2-R_4: Unused

User Instructions.

1. Key in the program and set End mode.
2. Key in the total number of coupons and press [n].
3. Key in the dollar amount of each periodic coupon and press [PMT].
4. Key in the periodic yield to maturity as a percent and press [i].
5. Key in the redemption value and press [FV].
6. Press [R/S] to calculate the number of periods in the duration.
7. For a new case, return to step 2.

Example. Calculate the duration of the following bond: $60 coupon, paid semi-annually for 5 years, 13.13% annual yield, $1000 redemption value.

Keys:	**Display:**	**Description:**
[9] [END]		Sets End mode.
5 [ENTER] 2 [×] [n]	10.00	Stores total number of coupons.
60 [PMT]	60.00	Stores amount of each periodic coupon.
13.13 [ENTER] 2 [÷] [i]	6.57	Stores semi-annual yield.
1000 [FV]	1,000.00	Stores redemption value.
[R/S]	7.75	Calculates duration in semi-annual periods.
2 [÷]	3.87	Duration in years.

Reference: Jess H. Chua, *A Closed-Form Formula for Calculating Bond Duration*, Financial Analysts Journal, May-June 1984.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-8

Actuarial Calculations

This collection of programs calculates the compound interest functions *n*, *i*, *PV*, *PMT*, and *FV* using the actuarial method. This means that the value for *n* can be any positive number.* The results duplicate those produced by the HP-37E, HP-38E/C, and HP-18C calculators.

The first program calculates *n*, *PV*, *PMT*, or *FV*. The second program, a shorter version of the first program, only calculates *n*. The third program calculates *i*.

Program to Calculate n, PV, PMT, FV.

Keys	Display
f P/R	
f PRGM	00-
STO 4	01- 44 4
1	02- 1
RCL i	03- 45 12
%	04- 25
STO 2	05- 44 2
+	06- 40
STO 1	07- 44 1
RCL 0	08- 45 0
y^x	09- 21
RCL PMT	10- 45 14
×	11- 20
STO 5	12- 44 5
1	13- 1
RCL 4	14- 45 4
g x≤y	15- 43 34
g GTO 69	16-43,33 69
x≷y	17- 34
RCL 1	18- 45 1
RCL n	19- 45 11
CHS	20- 16
y^x	21- 21

Keys	Display
STO 3	22- 44 3
−	23- 30
RCL 2	24- 45 2
÷	25- 10
STO 4	26- 44 4
RCL 5	27- 45 5
×	28- 20
STO 6	29- 44 6
x≷y	30- 34
2	31- 2
x≷y	32- 34
g x≤y	33- 43 34
g GTO 61	34-43,33 61
3	35- 3
x≷y	36- 34
g x≤y	37- 43 34
g GTO 47	38-43,33 47
RCL 6	39- 45 6
RCL PV	40- 45 13
+	41- 40
RCL 3	42- 45 3
÷	43- 10
CHS	44- 16

Keys	Display
STO FV	45- 44 15
g GTO 00	46-43,33 00
RCL FV	47- 45 15
RCL 3	48- 45 3
×	49- 20
RCL PV	50- 45 13
+	51- 40
RCL 4	52- 45 4
÷	53- 10
RCL 1	54- 45 1
RCL 0	55- 45 0
y^x	56- 21
÷	57- 10
CHS	58- 16
STO PMT	59- 44 14
g GTO 00	60-43,33 00
RCL FV	61- 45 15
RCL 3	62- 45 3
×	63- 20
RCL 6	64- 45 6
+	65- 40
CHS	66- 16
STO PV	67- 44 13

* Calculations using a non-integer value for n produce a mathematically correct result, but this result has no simple useful interpretation.

Keys	Display
g GTO 00	68-43,33 00
RCL 5	69- 45 5
RCL FV	70- 45 15
RCL 2	71- 45 2
x	72- 20
−	73- 30

Keys	Display
RCL 5	74- 45 5
RCL PV	75- 45 13
RCL 2	76- 45 2
x	77- 20
+	78- 40
÷	79- 10

Keys	Display
g LN	80- 43 23
RCL 1	81- 45 1
g LN	82- 43 23
÷	83- 10
STO n	84- 44 11
f P/R	

Registers:

n: Used
i: Used
PV: Used
PMT: Used
FV: Used
R_0: 0 or 1
R_1: $1+i$
R_2: i
R_3: Used
R_4: Used
R_5: Used
R_6: Used
R_7: Unused

User Instructions for n, PV, PMT, FV Program.

1. Key in the *n, PV, PMT, FV* program.
2. Press f FIN.
3. Store 0 in register 0 for End mode; store 1 in register 0 for Begin mode.
4. Key in the periodic interest rate; press i.
5. Key in any three of the following variables. Use the cash flow sign convention (money received is positive; money paid is negative).
   - Key in the total number of periods; press n.
   - Key in the present value; press PV.
   - Key in the periodic payment; press PMT.
   - Key in the future value; press FV.
6. To calculate *n*, press 1 R/S.
7. To calculate *PV*, press 2 R/S.
8. To calculate *PMT*, press 3 R/S.
9. To calculate *FV*, press 4 R/S.
10. For subsequent problems, return to step 2 and change values as needed.

Example. A $1000 loan has monthly payments of $80. If the annual interest rate is 9%, how many payments are necessary to amortize the loan? The payments are made at the end of each period.

Key in the *n, PV, PMT, FV* program.

Keys:	Display:	Description:
f FIN		Clears finance registers.
0 STO 0	0.00	Sets End mode.
1000 CHS PV	-1000.00	Stores known values.
9 g 12÷	0.75	
80 PMT	80.00	
0 FV	0.00	
1 R/S	13.17	Calculates number of monthly payments.

Program to Calculate n.

Keys	Display
f P/R	
f PRGM	00-
1	01- 1
RCL i	02- 45 12
%	03- 25
STO 2	04- 44 2
+	05- 40
STO 1	06- 44 1
RCL 0	07- 45 0
y^x	08- 21

Keys	Display
RCL PMT	09- 45 14
x	10- 20
STO 3	11- 44 3
RCL FV	12- 45 15
RCL 2	13- 45 2
x	14- 20
−	15- 30
RCL PV	16- 45 13
RCL 2	17- 45 2
x	18- 20

Keys	Display
RCL 3	19- 45 3
+	20- 40
÷	21- 10
g LN	22- 43 23
RCL 1	23- 45 1
g LN	24- 43 23
÷	25- 10
STO n	26- 44 11
f P/R	

Registers:

n: Used
i: Used
PV: Used
PMT: Used
FV: Used
R_0: 0 or 1
R_1: $1+i$
R_2: i
R_3: Used
R_5-$R_{.4}$ Unused

User Instructions for n Program.

1. Key in the n program.
2. Press f FIN.
3. Store 0 in register 0 for End mode; store 1 in register 0 for Begin mode.
4. Key in the periodic interest rate; press i.
5. Key in two or three of the following variables. Use the cash flow sign convention (money received is positive; money paid is negative).
   - Key in the present value; press PV.
   - Key in the periodic payment; press PMT.
   - Key in the future value; press FV.
6. Press R/S to calculate n, the total number of periods.
7. For subsequent problems, return to step 2 and change values as needed.

Example. You deposit $150.00 each month in an account paying 6½%, compounded monthly. How long will it take to accumulate $20,000.00? Assume End mode.

Key in the n program.

Keys:	Display:	Description:
f FIN		Clears finance registers.
0 STO 0	0.00	Sets a End mode.
150 CHS PMT	-150.00	Stores known values.
6.5 g 12÷	0.54	
20000 FV	20,000.00	
R/S	100.63	Calculates number of monthly payments.

Program to Calculate i.

Keys	Display
f P/R	
f PRGM	00-
.	01- 48
0	02- 0
1	03- 1
STO 3	04- 44 3
1	05- 1
RCL 3	06- 45 3
1	07- 1
+	08- 40
STO 1	09- 44 1
RCL n	10- 45 11
CHS	11- 16
y^x	12- 21
STO 2	13- 44 2
−	14- 30
RCL 3	15- 45 3
÷	16- 10
RCL 3	17- 45 3
RCL 0	18- 45 0
×	19- 20
1	20- 1
+	21- 40
×	22- 20

Keys	Display
RCL PMT	23- 45 14
×	24- 20
RCL PV	25- 45 13
+	26- 40
RCL FV	27- 45 15
RCL 2	28- 45 2
×	29- 20
+	30- 40
RCL 2	31- 45 2
RCL 1	32- 45 1
÷	33- 10
RCL n	34- 45 11
×	35- 20
STO 4	36- 44 4
1	37- 1
RCL 2	38- 45 2
−	39- 30
RCL 3	40- 45 3
÷	41- 10
−	42- 30
RCL PMT	43- 45 14
RCL 3	44- 45 3
÷	45- 10
×	46- 20

Keys	Display
RCL FV	47- 45 15
RCL 4	48- 45 4
×	49- 20
−	50- 30
÷	51- 10
STO − 3	52-44 30 3
ENTER	53- 36
CHS	54- 16
g x≤y	55- 43 34
x≷y	56- 34
EEX	57- 26
CHS	58- 16
8	59- 8
g x≤y	60- 43 34
g GTO 05	61-43,33 05
RCL 3	62- 45 3
EEX	63- 26
2	64- 2
×	65- 20
ENTER	66- 36
STO i	67- 44 12
f P/R	

Registers:

n: Used
i: Used
PV: Used
PMT: Used
FV: Used
R_0: 0 or 1
R_1: $1+i$
R_2: $(1+i)^{-n}$
R_3: i guess
R_4: Used
R_5-R_0: Unused

User Instructions for i Program.

1. Key in the i program.
2. Press f FIN.
3. Store 0 in register 0 for End mode; store 1 in register 0 for Begin mode.
4. Key in three or four of the following variables. Use the cash flow sign convention (money received is positive; money paid is negative).
   - Key in the total number of periods; press n.
   - Key in the present value; press PV.
   - Key in the periodic payment; press PMT.
   - Key in the future value; press FV.
5. Press R/S to calculate the periodic interest rate.
6. For subsequent problems, return to step 2 and change values as needed.

Example. You currently have $1000 in a savings account. If you wish to double your money in 5½ years, what annual interest rate do you need to earn?

Key in the *i* program.

Keys:	**Display:**	**Description:**
f FIN		Clears finance registers.
0 STO 0	0.00	Sets End mode
1000 CHS PV	-1000.00	Stores known values.
2000 FV	2000.00	
5.5 n	5.50	
0 PMT	0.00	
R/S	13.43	Calculates annual interest rate.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-9

Increasing Annuities

The following routines can be used to calculate the present and future values of an annuity that increases at a constant rate at equal intervals of time. Routines are included for both END and BEGIN mode calculations.

Present Value of an Increasing Annuity (END Mode).

1. Press [f][FIN] and [g][END].
2. Key in the total number of payment periods and press [n].
3. Key in the payment percentage increase per period expressed as one plus the decimal interest rate and press [ENTER]. (If there is a percentage decrease, key it in as one minus the decimal interest rate.)
4. Key in the discount (interest) rate per period expressed as one plus the decimal interest rate and press [Δ%] [i].
5. Key in the amount of the starting payment and press [x≷y] [÷] [PMT].
6. Press [PV] to calculate the present value of the payment stream.

Present Value of an Increasing Annuity (BEGIN Mode).

1. Press [f][FIN] and [g][END].
2. Key in the total number of payment periods and press [n].
3. Key in the discount (interest) rate per period expressed as one plus the decimal interest rate and press [ENTER].
4. Key in the payment percentage increase per period expressed as one plus the decimal interest rate and press [Δ%] [i].
5. Key in the amount of the starting payment and press [PMT].
6. Press [FV] to calculate the present value of the payment stream.

Future Value of an Increasing Annuity (END Mode).

1. Press [f][FIN] and [g][END].
2. Key in the total number of payment periods and press [n].
3. Key in the payment percentage increase per period expressed as one plus the decimal interest rate and press [ENTER]. (If there is a percentage decrease, key it in as one minus the decimal interest rate.)
4. Key in the discount (interest) rate per period expressed as one plus the decimal interest rate and press [Δ%] [i].
5. Key in the amount of the starting payment and press [x≷y] [÷] [PMT].
6. Press [PV] 0 [PMT].
7. Key in the discount (interest) rate as a percentage and press [i].
8. Press [FV] to calculate the future value of the payment stream.

Future Value of an Increasing Annuity (BEGIN Mode).

1. Press [f][FIN] and [g][END].
2. Key in the total number of payment periods and press [n].
3. Key in the discount (interest) rate per period expressed as one plus the decimal interest rate and press [ENTER].
4. Key in the payment percentage increase per period expressed as one plus the decimal interest rate and press [Δ%] [i].
5. Key in the amount of the starting payment and press [PMT].
6. Press [FV] [STO] [PV] 0 [PMT].
7. Key in the periodic discount (interest) rate as a percentage and press [i].
8. Press [FV] to calculate the future value of the payment stream.

Example 1. You are appraising a piece of income property that is providing increasing rents. Assuming a 7% rate of increase over the next 5 years, what is the present value of the income stream? Your discount rate is 12%, rent for the first year is expected to be $8,500, and payments occur at the end of the year.

Keys:	**Display:**	**Description:**
[f][FIN] [g][END]		
5 [n]		
1.07 [ENTER]		
1.12 [Δ%] [i]		
8500 [x≷y] [÷] [PMT]		
[PV]	-34,706.26	Calculates present value.

Example 2. Today you deposit $1000 into a savings account that earns 9 ½ interest, compounded annually. Each year you plan to increase the amount of your deposit by 15%. How much will you accumulate in 20 years?

Keys:	**Display:**	**Description:**
f FIN g END		
20 n		
1.095 ENTER		
1.15 Δ% i		
1000 PMT		
FV STO PV		
0 PMT		
9.5 i FV	203,568.97	Calculates future value.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-10

Annual Coupon Bond Price and Yield

This program calculates the price, accrued interest, and yield of an annual coupon bond. The bond can be either a short or long-term bond. Redemption can be at maturity or at call, the price is quoted as a percentage, and the calendar basis is Actual/Actual. For annual coupon bonds quoted on a 30/360 day basis, insert [R↓] after [g] [ΔDYS] at step 2, and change program lines 15, 30, 36, and 49 to the following:

Keys	Display
[g] [GTO] 48	**15-43,33 48**
[g] [GTO] 38	**30-43,33 38**
[g] [GTO] 67	**36-43,33 67**
[g] [GTO] 70	**49-43,33 70**

Program.

Keys	Display
[f] [P/R]	
[f] [PRGM]	**00-**
[g] [BEG]	**01- 43 7**
[g] [ΔDYS]	**02- 43 26**
[RCL] 5	**03- 45 5**
[÷]	**04- 10**
[STO] 6	**05- 44 6**
1	**06- 1**
[x≷y]	**07- 34**
[−]	**08- 30**
[RCL] 2	**09- 45 2**
[×]	**10- 20**
[STO] 7	**11- 44 7**
1	**12- 1**
[RCL] 3	**13- 45 3**
[g] [x≤y]	**14- 43 34**
[g] [GTO] 47	**15-43,33 47**

Keys	Display
[RCL] 2	**16- 45 2**
[PMT]	**17- 14**
[RCL] 2	**18- 45 2**
[RCL] 4	**19- 45 4**
[+]	**20- 40**
[FV]	**21- 15**
[RCL] 3	**22- 45 3**
1	**23- 1**
[−]	**24- 30**
[RCL] 6	**25- 45 6**
[+]	**26- 40**
[n]	**27- 11**
[RCL] 0	**28- 45 0**
[g] [x=0]	**29- 43 35**
[g] [GTO] 37	**30-43,33 37**
[i]	**31- 12**
[PV]	**32- 13**

Keys	Display
[CHS]	**33- 16**
[RCL] 7	**34- 45 7**
[−]	**35- 30**
[g] [GTO] 66	**36-43,33 66**
[RCL] [PMT]	**37- 45 14**
[RCL] 7	**38- 45 7**
[g] [x=0]	**39- 43 35**
[+]	**40- 40**
[RCL] 1	**41- 45 1**
[+]	**42- 40**
[CHS]	**43- 16**
[PV]	**44- 13**
[i]	**45- 12**
[g] [GTO] 00	**46-43,33 00**
[RCL] 0	**47- 45 0**
[g] [x=0]	**48- 43 35**
[g] [GTO] 69	**49-43,33 69**

Keys	Display
RCL 6	50- 45 6
RCL 0	51- 45 0
x	52- 20
EEX	53- 26
2	54- 2
+	55- 40
RCL 4	56- 45 4
RCL 2	57- 45 2
+	58- 40
EEX	59- 26
2	60- 2
x	61- 20

Keys	Display
x≷y	62- 34
÷	63- 10
RCL 7	64- 45 7
−	65- 30
g LSTx	66- 43 36
x≷y	67- 34
g GTO 00	68-43,33 00
RCL 4	69- 45 4
RCL 2	70- 45 2
+	71- 40
RCL 7	72- 45 7
RCL 1	73- 45 1

Keys	Display
+	74- 40
−	75- 30
g LSTx	76- 43 36
÷	77- 10
RCL 6	78- 45 6
÷	79- 10
EEX	80- 26
2	81- 2
x	82- 20
g GTO 00	83-43, 33 00
f P/R	

Registers:

n: Used
i: Used
PV: Used
PMT: Used
FV: Used
R_0: Yield
R_1: Price
R_2: Coupon
R_3: # Coupons
R_4: Redemption
R_5: # Days/Yr
R_6: Used
R_7: Used
R_8: Unused

User Instructions.

1. Key in the program, press f FIN to clear the finance registers, and press STO EEX if the C annunciator is not displayed.
2. Key in the amount of the annual coupon (as a percent) and press STO 2.
3. Key in the total number of coupons that are received and press STO 3.
4. Key in the redemption value and press STO 4.
5. Key in the number of days in a year (either 360 or 365) and press STO 5.
6. Key in the purchase price (if it is known) as a percentage of par and press STO 1.
7. Key in the annual yield (if it is known) as a percentage and press STO 0. If you wish to calculate the annual yield, press 0 STO 0.
8. Key in the settlement (purchase) date and press ENTER.
9. Key in the date of the next coupon and press R/S. If the annual yield is nonzero, the price is calculated. Otherwise, the annual yield is calculated.
10. When price is calculated, press x≷y to display the accrued interest.
11. For a new case, return to step 2. Only those values that have been changed need to be restored.

Example 1: Bond Price. What is the price and accrued interest of a 20-year Eurobond with annual coupons of 6.5%, purchased on August 15, 1986 to yield 7%? The next coupon is received on December 1, 1986. The calendar basis is Actual/Actual.

Keys:	Display:	Description:
[f] [FIN]		Clears financial registers.
[STO] [EEX]		Sets C annunciator if it is not lit.
6.5 [STO] 2	6.50	Stores annual coupon.
20 [STO] 3	20.00	Stores number of coupons.
100 [STO] 4	100.00	Stores redemption value.
365 [STO] 5	365.00	Stores number of days/year.
7 [STO] 0	7.00	Stores annual yield.
8.151986 [ENTER]	8.15	Enters settlement date.
12.011986 [R/S]	94.75	Enters next coupon date and calculates purchase price.
[x≷y]	4.58	Displays accrued interest.

Example 2: Bond Yield. What is the yield on a 15-year annual coupon bond purchased on September 15, 1986 at a price of 87½? The next coupon of 7.6% will be received on April 15, 1987 and the calendar basis is Actual/Actual.

Keys:	Display:	Description:
[f] [FIN]		Clears financial registers.
[STO] [EEX]		Sets C annunciator if it is not lit.
7.6 [STO] 2	7.60	Stores known values.
15 [STO] 3	15.00	
100 [STO] 4	100.00	
365 [STO] 5	365.00	
87.5 [STO] 1	87.50	
0 [STO] 0	0.00	
9.151986 [ENTER]	9.15	Enters settlement date.
4.151987 [R/S]	9.18	Calculates annual yield.

Example 3: Bond Price (Short Term). Calculate the price and accrued interest of the following annual coupon bond: 7% coupon, 8% yield, $100 redemption, settlement date March 21, 1986, maturity date December 1, 1986, actual day calendar. (The maturity date is the next coupon date and one coupon will be paid.)

Keys:	Display:	Description:
[f] [FIN]		Clears financial registers.
[STO] [EEX]		Sets C annunciator if it is not lit.
7 [STO] 2	7.00	Stores known values.
1 [STO] 3	1.00	
100 [STO] 4	100.00	
365 [STO] 5	365.00	
8 [STO] 0	8.00	
3.211986 [ENTER]	3.21	
12.011986 [R/S]	99.23	Calculates price.
[x≷y]	2.11	Displays accrued interest.

HEWLETT PACKARD

HP-12C Solutions 12-11

Deposits Needed to Meet a Future Cash Flow Need

Sometimes you want to know how much money you need to save now to accumulate money for a future series of outflows. An example of this situation is saving money for college. The following procedure helps determine how much you need to save each period. You need to know when you need the money, how much is needed, and at what interest rate you can invest.

User Instructions.

Part 1. Calculate the present value of the future withdrawals using NPV. Assume a cash flow of 0 at each period where there is no withdrawal.

1. Press [f] [REG] [g] [END].
2. Press 0 [g] [CFo].
3. Press 0 [g] [CFj].
4. Key in the number of payment periods until the withdrawals begin and press [g] [Nj]. (If this number is greater than 99, break the number of payments into two (or more) groups.)
5. Key in the withdrawal amount and press [g] [CFj].
6. Using [g] [CFj] and [g] [Nj], continue entering cash flows of 0 and the withdrawal amount through the last withdrawal.
7. Key in the periodic interest rate and press [i].
8. Press [f] [NPV] to calculate the net present value of the future cash flows.

Part 2. Solve for the periodic deposit necessary over the entire term.

1. Key in the total number of deposits and press [n].
2. Press [PMT] to calculate the periodic payment amount.

Example. Your daughter will be going to college in 12 years and you are starting a fund for her education. She will need $15,000 at the beginning of each year for four years. The fund earns 9%, compounded monthly, and you plan to make monthly deposits, starting at the end of the current month. The cash flow diagram looks like this:



How much should you deposit each month to meet her educational expenses?

Keys:	**Display:**	**Description:**
[f] [REG] [g] [END]		Clears finance registers and sets END mode.
0 [g] [CFo]	0.00	Stores initial cash flow.
0 [g] [CFj] 99 [g] [Nj]	0.00	Stores cash flows until withdrawals begin.
0 [g] [CFj] 44 [g] [Nj]	0.00	
15000 [g] [CFj]	15,000.00	Stores first withdrawal.
0 [g] [CFj] 11 [g] [Nj]	0.00	Stores cash flow of 0.
15000 [g] [CFj]	15,000.00	Stores second withdrawal.
0 [g] [CFj] 11 [g] [Nj]	0.00	Stores cash flow of 0.
15000 [g] [CFj] 0 [g] [CFj] 11 [g] [Nj] 15000 [g] [CFj]	15,000.00	Stores subsequent cash flows.
9 [g] [12÷]	0.75	Stores monthly interest rate.
[f] [NPV]	17,973.48	Calculates NPV of withdrawals.
15 [g] [12×]	180.00	Stores total number of deposits.
[PMT]	-182.30	Calculates monthly payment.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

HEWLETT PACKARD

HP-12C Solutions 12-12

Price and Yield to Call (Actual/Actual Day Basis)

This program calculates price and yield to call, assuming a semi-annual coupon payment and an actual/actual day basis. Prices are based on a par value of 100.

Program.

Keys	Display
f P/R	
f PRGM	00-
STO 2	01- 44 2
x≷y	02- 34
STO 1	03- 44 1
RCL PV	04- 45 13
RCL FV	05- 45 15
EEX	06- 26
2	07- 2
÷	08- 10

Keys	Display
STO 0	09- 44 0
÷	10- 10
PV	11- 13
RCL PMT	12- 45 14
RCL 0	13- 45 0
÷	14- 10
PMT	15- 14
RCL 1	16- 45 1
RCL 2	17- 45 2
RCL i	18- 45 12

Keys	Display
g x=0	19- 43 35
g GTO 26	20-43, 33 26
R↓	21- 33
f PRICE	22- 42 21
RCL 0	23- 45 0
x	24- 20
g GTO 00	25-43, 33 00
R↓	26- 33
f YTM	27- 42 22
f P/R	

Registers

n: Used
i: Used
PV: Used
PMT: Used
FV: Used
R_0: Used
R_1: Settlement
R_2: Maturity
R_3-R_6: Unused

User Instructions.

1. Key in the program.
2. Key in the annual coupon rate (as a percentage) and press PMT.
3. Key in the call price and press FV.
4. Key in the purchase price and press PV. If the purchase price is unknown, key in 0.
5. Key in the annual yield (as a percentage) and press i. If the yield is unknown, key in 0.
6. Key in the settlement date and press ENTER.
7. Key in the call date and press R/S. If the purchase price is 0, the price to call is calculated. If the yield is 0, the yield to call is calculated.
8. For a new problem, return to step 2.

Example 1. A 10% coupon bond was purchased October 14, 1980 for 97.25. If the bond is called on March 16, 1986 for 103, what is the yield to call?

Keys:	Display:	Description:
10 [PMT]	10.00	Stores annual coupon rate.
103 [FV]	103.00	Stores call price.
97.25 [PV]	97.25	Stores price.
0 [i]	0.00	Stores 0 in i.
10.141980 [ENTER]	10.14	Stores settlement date.
3.161986 [R/S]	11.10	Calculates yield to call.

Example 2. A bond with a 9.5% coupon and a call of 102 has a settlement date of August 28, 1981 and a call date of June 1, 1990. If the bond is to yield 11%, what is the purchase price?

Keys:	Display:	Description:
9.5 [PMT]	9.50	Stores annual coupon rate.
102 [FV]	102.00	Stores call price.
11 [i]	11.00	Stores yield to call.
8.281981 [ENTER]	8.28	Stores settlement date.
6.011990 [R/S]	92.45	Calculates purchase price.

HEWLETT PACKARD

Portable Computer Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330

12-12
4/87